

Voorgeslacht rapport generator

versie 3.5 (16 mei 2006)

auteur: Berend Meijer

Beschrijving

De *voorgeslacht rapport generator* is een sjabloon voor gebruik met het genealogie-programma Aldfaer (www.aldfaer.nl). Met dit sjabloon kunnen kwartierstaten, stamreeksen en matrilineaire reeksen aangemaakt worden. Indien gewenst kan het sjabloon diverse indexen en lijsten aanmaken en aan het eind van het document invoegen.

De uitvoer van het sjabloon is een HTML pagina, met de naam *kwartierstaat.html*, die in een willekeurige web browser geopend en bekeken kan worden. Het is mogelijk de aangemaakte pagina te optimaliseren voor verwerking met Microsoft Word. In dit geval zijn slechts enkele bewerkingen nodig om de HTML pagina om te zetten in een publiceerbaar document.



Het sjabloon ondersteunt vrijwel alle Aldfaer instellingen (te vinden onder *Instellingen...* in het *Bestand* menu), met uitzondering van de privacy-instellingen. Daarnaast zijn er nog een aantal specifieke instellingen die bij de uitvoering van het sjabloon ingesteld kunnen worden. Tenslotte zijn er nog enkele minder vaak gebruikte instellingen die in de tekst van het sjabloon zelf gewijzigd kunnen worden.

Aldfaer instellingen

Algemene instellingen

De rood omcirkelde instellingen worden door het sjabloon ondersteund. Zie de Aldfaer documentatie voor de betekenis en werking van deze instellingen.

Niet elk lettertype geeft bevredigende resultaten. In de uitvoer van het sjabloon worden een aantal symbolen gebruikt, die niet door alle lettertypes ondersteund worden,




zoals de mannelijke en vrouwelijke geslachtssymbolen. Times New Roman en Arial zijn goed bruikbare lettertypes.

Presentatie instellingen

De rood omliggende instellingen worden door het sjabloon ondersteund. De betekenis en werking van deze instellingen zijn te vinden in de Aldfaer documentatie.

Sjabloon-instellingen


Bij de uitvoering van het sjabloon worden één of meer dialogen getoond, waarmee de werking van het sjabloon gewijzigd kan worden.



Hoofddialoog

De belangrijkste sjabloon-instellingen kunnen met de hoofddialoog ingesteld worden. Niet alle hier getoonde instellingen zijn altijd beschikbaar. Een aantal instellingen zijn uitbreidingen van de eerder genoemde Aldfaer instellingen; dezen zijn alleen beschikbaar als de betreffende Aldfaer instellingen aan staan.



- *Optimaal voor MS Word.* Als deze instelling aan staat, dan wordt de uitvoer geoptimaliseerd voor verdere bewerking met Microsoft Word. Als deze instelling uit staat, dan is de uitvoer het meest geschikt om te bekijken in een web browser.
- *MS Word: maak een titelpagina.* (Deze instelling heeft alleen effect als de *optimaal voor MS Word* instelling aan staat.) Als de instelling aan staat, dan wordt een titelpagina ingevoegd aan het begin van het document.¹ Als deze instelling uit staat, dan wordt geen titelpagina ingevoegd.
- *MS Word: wijzig titelfoto.* (Deze instelling heeft alleen effect als zowel de *optimaal voor MS Word* instelling als de *MS Word: maak een titelpagina* instelling aan staan.) Als de instelling aan staat, dan kun je in een nieuwe dialoog kiezen welke foto op de titelpagina opgenomen wordt en hoe groot deze foto afgebeeld moet worden. Als deze instelling uit staat, dan worden een vaste foto en grootte gebruikt.
- *MS Word: nieuwe pagina voor elke generatie.* (Deze instelling heeft alleen effect als de *optimaal voor MS Word* instelling aan staat.) Als de instelling aan staat, dan begint de beschrijving van elke generatie op een nieuwe pagina. Indexen en dergelijke beginnen in dit geval ook steeds op een nieuwe pagina. Als deze instelling uit staat, dan worden geen nieuwe pagina's ingevoegd en is de tekst van het document doorlopend.
- *MS Word: onderdruk aanklikbare verwijzingen.* (Deze instelling heeft alleen effect als de *optimaal voor MS Word* instelling aan staat.) Als de instelling aan staat, dan worden er geen aanklikbare verwijzingen (oftewel hyperlinks) in de uitvoer opgenomen. Voor grote documenten is dit aan te raden, om problemen te vermijden bij het openen van het document in Microsoft Word. Als deze instelling uit staat dan zijn alle nummer-verwijzingen in het document aanklikbaar, net als wanneer de uitvoer niet voor Microsoft Word geoptimaliseerd wordt.
- *MS Word: verborgen kop/voet-tekst.* (Deze instelling heeft alleen effect als de *optimaal voor MS Word* instelling aan staat.) Als de instelling aan staat, dan bevat de uitvoer onzichtbare tekst die gebruikt kan worden voor het aanmaken van kop- of voetteksten in Microsoft Word.² Als deze instelling uit staat, dan bevat de uitvoer niet zulke onzichtbare tekst.



¹ Zie de verderop beschreven Word macro voor de beste manier om de titelpagina verder te bewerken.



² Zie de verderop beschreven Word macro voor de beste manier om deze onzichtbare tekst te gebruiken voor het aanmaken van voetteksten.

- *'Aanwezige' in plaats van 'getuige'*. (Deze instelling is alleen beschikbaar als Aldfaer's presentatie instelling *getuigen* aan staat.) Als de instelling aan staat, dan worden getuigen in de uitvoer aangegeven als 'aanwezigen'. Als deze instelling uit staat, dan worden getuigen aangegeven als 'getuigen'. De beste waarde voor deze instelling hangt af van welke personen je in Aldfaer als getuigen bij aangiftes en dergelijke opneemt.
- *Voeg index van personen in*. (Deze instelling is alleen beschikbaar als Aldfaer's presentatie instelling *index* aan staat.) Als de instelling aan staat, dan wordt een index van personen in de uitvoer opgenomen. Als deze instelling uit staat, dan wordt geen index van personen opgenomen.
- *Personen-index: sorteer op voornaam*. (Deze instelling heeft alleen effect als de *voeg index van personen in* instelling aan staat.) Als de instelling aan staat, dan wordt de index van personen op voornaam gesorteerd. Als deze instelling uit staat, dan wordt de index van personen op achternaam gesorteerd.
- *Voeg index van plaatsen in*. (Deze instelling is alleen beschikbaar als Aldfaer's presentatie instelling *index* aan staat.) Als de instelling aan staat, dan wordt een index van plaatsen in de uitvoer opgenomen. Als deze instelling uit staat, dan wordt geen index van plaatsen opgenomen.
- *Voeg index van beroepen en functies in*. (Deze instelling is alleen beschikbaar als Aldfaer's presentatie instelling *index* aan staat. De instelling heeft geen effect als Aldfaer's presentatie instellingen *beroep* en *overige feiten* uit staan.) Als de instelling aan staat, dan wordt een index van beroepen en functies in de uitvoer opgenomen. Als deze instelling uit staat, dan wordt geen index van beroepen en functies opgenomen.
- *Voeg lijst van afbeeldingen in*. (Deze instelling is alleen beschikbaar als Aldfaer's presentatie instelling *index* aan staat. De instelling heeft geen effect als Aldfaer's presentatie instelling *foto* uit staat.) Als de instelling aan staat, dan wordt een lijst van afbeeldingen in de uitvoer opgenomen. Als deze instelling uit staat, dan wordt geen lijst van afbeeldingen opgenomen.
- *Voeg bron-aantekeningen in*. (Deze instelling is alleen beschikbaar als Aldfaer's presentatie instelling *bronvermeldingen* aan staat.) Als de instelling aanstaat, dan worden alle bron-aantekeningen³ in de uitvoer opgenomen, volgend op de lijst van bronnen. Als deze instelling uit staat, dan worden geen bron-aantekeningen opgenomen.
- *Wijzig foto-instellingen*. (Deze instelling is alleen beschikbaar als Aldfaer's presentatie instelling *foto* aan staat.) Als de instelling aanstaat, dan kun je in een nieuwe dialoog het maximum aantal foto-rijen per persoon instellen. Als deze instelling uit staat, dan word een vast aantal foto-rijen van een vaste breedte gebruikt.
- *Wijzig laatst mogelijke Juliaanse jaar*. (Deze instelling is alleen beschikbaar als de Aldfaer's algemene instelling *naam van de dag* aan staat.) Als de instelling aan staat, dan kun je in een nieuwe dialoog het laatste jaar aangeven, waarin mogelijk de Juliaanse kalender gebruikt werd. Als deze instelling uit staat, dan wordt een vast jaar aangehouden. In de uitvoer van het sjabloon worden dagnamen alleen toegevoegd aan data na het aangegeven jaar.



³ Bron-aantekeningen zijn de aantekeningen die in Aldfaer gevonden worden in de sectie *Bronnen* onder *Aantekeningen...* in het *Extra* menu. Huidige Aldfaer versies geven geen nette opmaak van deze aantekeningen, zodat deze instelling het beste uitgezet kan worden.

- *Wijzig feit-categorieën.* (Deze instelling is alleen beschikbaar als Aldfaer's presentatie instelling *overige feiten* aan staat.) Als de instelling aanstaat, dan kun je in een aantal nieuwe dialogen aangeven welke categorieën feiten (van het soort *overig*) niet in de uitvoer opgenomen moeten worden. Als deze instelling uit staat, dan worden de feiten van alle categorieën in de uitvoer opgenomen.



Titelfoto dialoog

Met deze dialoog kies je welke foto op de titelpagina opgenomen wordt en hoe groot deze foto afgebeeld moet worden.

- *Titelfoto: persoon (kwartiernummer).* Deze instelling bepaald van welke persoon de foto afkomstig is. De betreffende persoon wordt gekozen aan de hand van zijn/haar kwartiernummer (of reeksnummer in het geval van een stamreeks of matrilineaire reeks). Met de waarde nul wordt er geen foto op de titelpagina geplaatst.
- *Titelfoto: foto-nummer (0 = laatste).* Deze instelling bepaald welke foto van de gekozen persoon gebruikt wordt. De waarde nul heeft een speciale betekenis; met deze waarde wordt de laatste foto van de gekozen persoon gebruikt. Als de gekozen foto niet bestaat, dan wordt er geen foto op de titelpagina geplaatst.
- *Titelfoto: hoogte (in pixels).* Deze instelling bepaald de grootte van de foto op de titelpagina. Deze grootte moet worden opgegeven in pixels.





Foto-instellingen dialoog

Met deze dialoog kun je het maximale aantal foto-rijen kiezen dat bij elke persoon opgenomen wordt. Het aantal rijen, de rij-breedte en de breedtes van de individuele foto's bepalen het maximale aantal foto's.⁴

- *Maximaal aantal foto-rijen per persoon (minstens 1).* Deze instelling bepaald het maximale aantal foto-rijen voor elke persoon. De waarde nul kan niet gebruikt worden.
- *Maximale breedte van een foto-rij (in pixels).* Deze instelling bepaald de breedte van een foto-rij. Deze breedte moet worden opgegeven in pixels. Een te lage waarde leidt ertoe dat niet de volle breedte van het document gebruikt wordt. Een te hoge waarde leidt ertoe dat sommige foto's (deels) buiten de marges van het document vallen.⁵



Laatste Juliaanse jaar dialoog


Met deze dialoog kun je het laatste jaar instellen, waarvoor mogelijk de Juliaanse kalender van toepassing is. De Juliaanse kalender werd gebruikt voordat de huidige (Gregoriaanse) kalender ingevoerd werd. Aldfaer's berekening van wekdagen is gebaseerd op de Gregoriaanse kalender en geeft verkeerde resultaten voor data die gebaseerd zijn op de Juliaanse kalender. De Gregoriaanse

⁴ De breedte en hoogte van individuele foto's wordt in Aldfaer opgegeven met de algemene instellingen *breedte* en *hoogte* onder *afbeelding in pixels*.

⁵ Een rij-breedte van 580 pixels is optimaal als de verderop beschreven Word macro gebruikt wordt voor het verder bewerken van het document. In het algemeen is dit ook een goede waarde voor documenten die op A4 papier afgedrukt worden.

kalender is in verschillende delen van Nederland op verschillende tijdstippen ingevoerd, maar voor zover mij bekend nergens later dan in de loop van 1701.

- *Laatst mogelijke Juliaanse jaar.* In de uitvoer van het sjabloon worden dagnamen alleen toegevoegd aan data na het hier ingestelde jaar.



Feit-categorie dialogen

Met deze dialogen kun je instellen of feiten (van het soort *overig*) van een bepaalde categorie wel of niet in de uitvoer opgenomen moeten worden. Elke keer dat bij de uitvoering van het sjabloon een nieuwe feit-categorie tegengekomen wordt, word je gevraagd of feiten van deze categorie al of niet verborgen moeten worden. In tegenstelling tot de eerder genoemde dialogen, die bij het starten van het sjabloon getoond worden, worden deze dialogen pas gedurende de uitvoering van het sjabloon getoond.

- *Verberg feit-categorie 'divers: xxxxx'.* Deze instelling bepaald of feiten van de genoemde categorie al of niet in de uitvoer opgenomen worden.

Interne instellingen

In de tekst van het sjabloon zelf kunnen een aantal minder vaak gebruikte instellingen gewijzigd worden. Al deze instellingen zijn te vinden aan het begin van het sjabloon. Een beschrijving van elke instelling wordt daar ook gegeven.

Sommige interne instellingen (met namen als *F_DialoogOptie_Xxxxx*) bepalen de in eerste instantie ingevulde waarden in de hiervoor beschreven dialogen. Met de interne instelling *F_Optie_GebruikDialoog* kun je desgewenst het gebruik van dialogen helemaal uitschakelen en altijd de ongewijzigde waarden van de interne instellingen gebruiken.

Andere interne instellingen bepalen bijvoorbeeld de kleur van diverse elementen in de uitvoer van het sjabloon.

Microsoft Word macro

Als de sjabloon-instelling *Optimaal voor MS Word* aan staat, dan wordt de uitvoer van het sjabloon geoptimaliseerd voor verwerking met Microsoft Word. De uitvoer van het sjabloon is echter altijd een HTML pagina, met beperkte opmaak-mogelijkheden. Daardoor zijn er zaken die niet direct vanuit het sjabloon geregeld kunnen worden.

Om die reden voegt het sjabloon een aantal voor de gebruiker onzichtbare eigenschappen toe, voornamelijk in de vorm van alinea-stijlen en verborgen tekst. Deze eigenschappen maken het relatief eenvoudig om in Microsoft Word de aangemaakte HTML pagina om te zetten in een voor publicatie geschikt document. De hierna volgende Word macro kan deze taak automatisch uitvoeren.

Deze macro is geschikt voor Engelstalige versies van Microsoft Word. Voor gebruik in een uitsluitend Nederlandstalige versie is vertaling noodzakelijk.

```

Sub AldfaerRapport()
'
' versie 3.2 (19 maart 2006)
'
' -----
' Negeer fouten tijdens de macro uitvoering. Deactiveer het scherm.
' -----
On Error Resume Next
Application.ScreenUpdating = False
' -----
' Schakel spelling en grammatica checks uit.
' -----
ActiveDocument.Range().NoProofing = True
' -----
' Zorg ervoor dat fotos en ondertekeningen altijd op dezelfde pagina staan.
' -----
For Each Table In ActiveDocument.Tables
  For Each Row In Table.Rows
    Row.AllowBreakAcrossPages = False
  Next
Next
' -----
' Zorg ervoor dat alle regels van een paragraaf altijd op dezelfde pagina
' staan. Voor erg lange paragrafen laten we echter Word's weduwen/wezen
' mechanisme zijn gang gaan.
' -----
With ActiveDocument.Styles(wdStyleNormal).ParagraphFormat
  .KeepTogether = True
  .WidowControl = True
  .KeepWithNext = False
End With
For Each Paragraph In ActiveDocument.Paragraphs
  If Paragraph.Range.Characters.Count > _
    5500 / ActiveDocument.Styles("standaard").Font.Size Then
    Paragraph.KeepTogether = False
  End If
Next
' -----
' Zorg ervoor dat een een alleenstaande "Kinderen van" regel altijd op
' dezelfde pagina staat als het eerste kind.
' -----
With ActiveDocument.Styles("kinderen").ParagraphFormat
  .KeepWithNext = True
End With
' -----
' Zorg ervoor dat een feittitel altijd op dezelfde pagina staat als het
' eerste bijbehorende feit.
' -----
With ActiveDocument.Styles("feittitel").ParagraphFormat
  .KeepWithNext = True
End With
' -----
' Zorg ervoor dat een titel altijd op dezelfde pagina staat als de paragraaf
' die op de titel volgt. Als een generatie-titel gevolgd wordt door een
' paragraaf met verborgen tekst, dan is de (hierna volgende) instelling van
' stijl "verberg" ook van belang.
' -----
With ActiveDocument.Styles(wdStyleHeading2).ParagraphFormat
  .KeepWithNext = True
End With
' -----
' Zorg ervoor dat de verborgen teksten voor de voetregels altijd op de juiste
' pagina staan. Als de verborgen tekst volgt op een titel, dan zorgen we
' ervoor dat de titel en de volgende paragraaf op dezelfde pagina staan.
' -----
For Each Paragraph In ActiveDocument.Paragraphs
  If Paragraph.Style = "verberg" Then
    If Paragraph.Previous(1).Style = wdStyleHeading2 Then
      Paragraph.KeepWithNext = True
    Else

```

```

        Paragraph.Previous(1).KeepWithNext = True
    End If
End If
Next
' -----
' Zet pagina formaat, marges en layout voor tweezijdig afdrucken.
' -----
With ActiveDocument.PageSetup
    .Orientation = wdOrientPortrait
    .PageWidth = CentimetersToPoints(21)
    .PageHeight = CentimetersToPoints(29.7)
    .TopMargin = CentimetersToPoints(2.5)
    .BottomMargin = CentimetersToPoints(3.5)
    .LeftMargin = CentimetersToPoints(2.5)
    .RightMargin = CentimetersToPoints(2.5)
    .Gutter = CentimetersToPoints(1.5)
    .HeaderDistance = CentimetersToPoints(1)
    .FooterDistance = CentimetersToPoints(2)
    .OddAndEvenPagesHeaderFooter = True
    ' Onderdruk de voettekst op de titelpagina, als die aanwezig is.
If ActiveDocument.Paragraphs(1).Style = "verberg" Then
    .DifferentFirstPageHeaderFooter = True
Else
    .DifferentFirstPageHeaderFooter = False
End If
    .VerticalAlignment = wdAlignVerticalTop
    .MirrorMargins = True
    .TwoPagesOnOne = False
    .GutterPos = wdGutterPosLeft
End With
' -----
' Voeg voetregels in.
' -----
' Lettertype en tabulaties.
With ActiveDocument.Styles(wdStyleFooter)
    .Font = ActiveDocument.Styles("standaard").Font
    .ParagraphFormat.TabStops.ClearAll
    .ParagraphFormat.TabStops.Add _
        Position:=CentimetersToPoints(14.5), Alignment:=wdAlignTabRight, _
        Leader:=wdTabLeaderSpaces
End With
' Oneven pagina's.
ActiveDocument.Sections(1).Footers(wdHeaderFooterPrimary).Range.Select
With Selection
    .Range.Fields.Add Range:=Selection.Range, Type:=wdFieldStyleRef, _
        Text:"verberg", PreserveFormatting:=True
    .TypeText Text:=vbTab
    .Range.Fields.Add Range:=Selection.Range, Type:=wdFieldPage, _
        Text:"", PreserveFormatting:=True
End With
' Even pagina's.
ActiveDocument.Sections(1).Footers(wdHeaderFooterEvenPages).Range.Select
With Selection
    .Range.Fields.Add Range:=Selection.Range, Type:=wdFieldPage, _
        Text:"", PreserveFormatting:=True
    .TypeText Text:=vbTab
    .Range.Fields.Add Range:=Selection.Range, Type:=wdFieldStyleRef, _
        Text:"verberg", PreserveFormatting:=True
End With
' -----
' Voeg een blanco pagina in na de titelpagina, indien aanwezig.
' -----
If ActiveDocument.Paragraphs(1).Style = "verberg" Then
    For Each Paragraph In ActiveDocument.Paragraphs
        ' Het lijkt alsof een fout in VB het onmogelijk maakt om hier de waarde
        ' wdStyleHeading2 te gebruiken.
        If Paragraph.Style = "Heading 2" Then
            ' Voor alle zekerheid testen we ook de stijl van de vorige paragraaf.
            If Paragraph.Previous(1).Style = "verberg" Then
                Paragraph.Previous(1).PageBreakBefore = True
            End If
        End If
    Next
End If

```

```
        End If
        Exit For
    End If
Next
End If
' -----
' Toon het document in opmaak-vorm.
' -----
With ActiveDocument.ActiveWindow
    If .View.SplitSpecial = wdPaneNone Then
        .ActivePane.View.Type = wdPrintView
    Else
        .View.Type = wdPrintView
    End If
End With
' -----
' Open het afdruk-voorbeeld.
' -----
ActiveDocument.PrintPreview
With ActiveDocument.ActiveWindow.ActivePane.View.Zoom
    .PageColumns = 1
    .PageRows = 1
End With
' -----
' Activeer het scherm.
' -----
Application.ScreenUpdating = True
End Sub
```